Source Code:

```
#include "LedControl.h"
int PinTiltX = 2; //Pin for Tiltsensor X
int PinTiltY = 3; //Pin for Tiltsensor Y
int PinCLK = 8;
int PinCS = 9;
int PinDIN = 10;
LedControl lc = LedControl(PinDIN, PinCLK, PinCS, 1);
int DicePic[8][6][2] =
{
  { //empty matrix and start position:
    {-1,-1}, //1. Point
    {-1,-1}, //2. Point
    {-1,-1}, //3. Point
    {-1,-1}, //4. Point
    {-1,-1}, //5. Point
    {-1,-1}, //6. Point
  },
  { //1:
    {4,4}, //1. Point
    {-1,-1}, //2. Point
    {-1,-1}, //3. Point
    {-1,-1}, //4. Point
    {-1,-1}, //5. Point
    {-1,-1}, //6. Point
  },
  { //2:
    {2,2}, //1. Point
    {6,6}, //2. Point
    {-1,-1}, //3. Point
    {-1,-1}, //4. Point
    {-1,-1}, //5. Point
    {-1,-1}, //6. Point
  },
  { //3:
    {2,6}, //1. Point
    {6,2}, //2. Point
    {4,4}, //3. Point
    {-1,-1}, //4. Point
    {-1,-1}, //5. Point
    {-1,-1}  //6. Point
```

```
  },
  { //4:
    {2,2}, //1. Point
    {2,6}, //2. Point
    {6,2}, //3. Point
    {6,6}, //4. Point
    {-1,-1}, //5. Point
    {-1,-1}, //6. Point
  },
  { //5:
    {2,2}, //1. Point
    {2,6}, //2. Point
    {6,2}, //3. Point
    {6,6}, //4. Point
    {4,4}, //5. Point
    {-1,-1}, //6. Point
  },
  { //6:
    {2,1}, //1. Point
    {2,4}, //2. Point
    {2,7}, //3. Point
    {6,1}, //4. Point
    {6,4}, //5. Point
    {6,7}  //6. Point
  },
  { //Start:
    {-1,-1}, //1. Point
    {-1,-1}, //2. Point
    {-1,-1}, //3. Point
    {-1,-1}, //4. Point
    {-1,-1}, //5. Point
    {-1,-1}  //6. Point
  }
  };
float DiceXpos[6];
float DiceXdir[6];
volatile byte DiceXspeed[6];
float DiceYpos[6];
float DiceYdir[6];
volatile byte DiceYspeed[6];

int DiceValue;
unsigned long timestamp;
```

```
byte Mode;
int volatile shakes;
int ShakesPerSecond;
int step;

void InterruptChecks() {
  shakes=shakes+1;
  Serial.println(millis());
  timestamp=millis();
}
void SetSpeedX() {
  if (Mode==0) {
    for (int i = 0; i < 6; i++) {
      if (DiceXspeed[i]<255) {DiceXspeed[i]=DiceXspeed[i]+5;}
    }
  }
  InterruptChecks();
}
void SetSpeedY() {
  if (Mode==0) {
    for (int i = 0; i < 6; i++) {
      if (DiceYspeed[i]<255) {DiceYspeed[i]=DiceYspeed[i]+5;}
    }
  }
  InterruptChecks();
}
void ShowLed(int x, int y, bool onoff) {
  if ((x<8) and (y<8) and (x>=0) and (y>=0)) {
    lc.setLed(0, x, y, onoff);
  }
}
void ShowDot(int x, int y, bool onoff) {
  ShowLed(x-1, y-1, onoff);
  ShowLed(x, y-1, onoff);
  ShowLed(x-1, y, onoff);
  ShowLed(x, y, onoff);
}
void ShowDicePic(int value) {
boolean done;
  for (int i = 0; i < 6; i++) {
    DiceXspeed[i]=100;
    DiceYspeed[i]=100;
```

```
    DiceXdir[i]=0;
    if (int(DiceXpos[i])>DicePic[value][i][0]) {DiceXdir[i]=-1;}
    else if (int(DiceXpos[i])<DicePic[value][i][0]) {DiceXdir[i]=1;}

    DiceYdir[i]=0;
    if (int(DiceYpos[i])>DicePic[value][i][1]) {DiceYdir[i]=-1;}
    else if (int(DiceYpos[i])<DicePic[value][i][1]) {DiceYdir[i]=1;}
  }
  Serial.println(value);
  Serial.println("Start moving");
  do {
    Serial.println("Moving");
    for (int i = 0; i < 6; i++) {
      if (int(DiceXpos[i])!=DicePic[value][i][0]) {
        DoStep(DiceXpos[i],DiceXdir[i],DiceXspeed[i],false);
      }
      if (int(DiceYpos[i])!=DicePic[value][i][1]) {
        DoStep(DiceYpos[i],DiceYdir[i],DiceYspeed[i],false);
      }
    }
    lc.clearDisplay(0);
    for (int i = 0; i < 6; i++) {
      ShowDot(int(DiceXpos[i]), int(DiceYpos[i]), true);
    }
    delay(50);
    done=true;
    for (int i = 0; i < 6; i++) {
      if (int(DiceXpos[i])!=DicePic[value][i][0]) {done=false;}
      if (int(DiceYpos[i])!=DicePic[value][i][1]) {done=false;}
    }
  } while (done==false);
  Serial.println("End moving");
  lc.clearDisplay(0);
  for (int i = 0; i < 6; i++) {
    ShowDot(DicePic[value][i][0],DicePic[value][i][1], true);
  }
}
void DoStep(float &pos, float &dir, volatile byte &sp, bool check) {
  pos=pos+float(sp)/255*dir;
  if (check==true) {
    if (pos>7) {
       pos=7;
       dir=dir*(-1);
```

```
      }
    if (pos<1) {
       pos=1;
       dir=dir*(-1);
     }
  }
    if (sp>0) {sp=sp-1;}
}
void MoveDots() {
  for (int i = 0; i < 6; i++) {
    //calc new coordinates
    DoStep(DiceXpos[i],DiceXdir[i],DiceXspeed[i],true);
    DoStep(DiceYpos[i],DiceYdir[i],DiceYspeed[i],true);
  }
    //show dice points
    lc.clearDisplay(0);
    for (int i = 0; i < 6; i++) {
      ShowDot(int(DiceXpos[i]), int(DiceYpos[i]), true);
    }
}
void setup() {
  //The MAX7219 is in power-saving mode on startup,
  //we have to do a wakeup call
  lc.shutdown(0, false);
  //Set the brightness to a medium values
  lc.setIntensity(0, 8);
  //and clear the display
  lc.clearDisplay(0);
  randomSeed(analogRead(0));
  DiceValue=0;
  for (int i = 0; i < 6; i++) {
    DiceXpos[i]=DicePic[7][i][0];
    DiceYpos[i]=DicePic[7][i][1];
    DiceXdir[i]=random(3)-1;
    DiceYdir[i]=random(3)-1;
    DiceXspeed[i]=random(126)+120;
    DiceYspeed[i]=random(126)+120;
    }
  //Setup the pins
  pinMode(PinTiltX, INPUT_PULLUP);
  pinMode(PinTiltY, INPUT_PULLUP);
  attachInterrupt(digitalPinToInterrupt(PinTiltX),SetSpeedX,CHANGE);
  attachInterrupt(digitalPinToInterrupt(PinTiltY),SetSpeedY,CHANGE);
```

```
  lc.clearDisplay(0);
  timestamp=millis();
  Mode=1;
  ShowDicePic(6);
  delay(1000);

  lc.clearDisplay(0);
  Mode=0;
  Serial.begin(9600);
  step=0;
  shakes=0;
}
void loop() {
  delay(50);
  step=step+1;
  if (step>20) {
    //1 sec is over
    step=0;
    ShakesPerSecond=shakes;
    shakes=0;
  }
  if (Mode==0) {
    MoveDots();
    if (millis()-timestamp>2000) {
      //there is no shaking since 2 sec
      Mode=1;
      DiceValue=random(6)+1;
      ShowDicePic(DiceValue);
    }
  }
  if (ShakesPerSecond>5) {
    //shaking again
    Mode=0;
  }
}
```

Reference:

- https://www.arduino.cc/
- https://playground.arduino.cc/